Maple - Seminar 2

EXAMPLE 1: Write a Maple procedure to calculate the distance between two skew lines. The skew lines are given by their points and directional vectors: *p*: A=[7,3,9], u=(1,2,-1); *q*: B=[3,1,1], v=(-7,2,3).

```
> restart;
> DistSkewL:=proc(A,u,B,v)
>            local Cross, NormCross;
>                  Cross:=linalg[crossprod](u,v);
>                  NormCross:=Cross/linalg[norm](Cross,2);
>                  eval(abs(linalg[dotprod]((A-B),NormCross)));
> end:
```

We can save the procedure to the file MapleSem.m for the later using

```
> save DistSkewL,"MapleSem.m":
> restart;
> read "MapleSem.m":
> M:=([7,3,9]); a:=([1,2,-1]); N:=([3,1,1]); b:=([-7,2,3]);
```

$$M := [7, 3, 9]$$

$$a := [1, 2, -1]$$

$$N := [3, 1, 1]$$

$$b := [-7, 2, 3]$$

```
> DistSkewL(M,a,N,b);
```

$$2\sqrt{21}$$

```
> print(DistSkewL);
```

```
proc (A, u, B, v)
local Cross, NormCross;
    Cross := linalg[ crossprod ](u, v);
    NormCross := Cross/linalg[ norm ](Cross, 2);
    eval( abs( linalg[ dotprod ](A – B, NormCross ) ) )
end proc
```

There are some visualization commands to help in the understanding of basic concepts of linear algebra in the Student[LinearAlgebra] subpackage.

```
> with(Student[LinearAlgebra]):
Warning, the protected name . has been redefined and unprotected

> CrossProductPlot(Vector(a),Vector(b));
```




```
> VectorSumPlot(Vector(a),Vector(b));
```



EXAMPLE 2: Solve a system of linear equations which is given in matrix form

```
> restart;
> A:=matrix([[2,-3,1],[1,2,-1],[2,1,1]]); B:=Vector([0,3,12]);
```

$$A := \begin{bmatrix} 2 & -3 & 1 \\ 1 & 2 & -1 \\ 2 & 1 & 1 \end{bmatrix}$$

$$B := \begin{bmatrix} 0 \\ 3 \\ 12 \end{bmatrix}$$

```
> with(linalg):
Warning, the protected names norm and trace have been redefined and
unprotected
```

Regularity

```
> det(A);
```

$$12$$

Frobenius' rule

```
> Aa:=augment(A,B);
```

$$Aa := \begin{bmatrix} 2 & -3 & 1 & 0 \\ 1 & 2 & -1 & 3 \\ 2 & 1 & 1 & 12 \end{bmatrix}$$

```
> gausselim(Aa);
```

$$\begin{bmatrix} 2 & -3 & 1 & 0 \\ 0 & 4 & 0 & 12 \\ 0 & 0 & \frac{-3}{2} & \frac{-15}{2} \end{bmatrix}$$

```
> gaussjord(Aa);
```

$$\begin{bmatrix} 1 & 0 & 0 & 2 \\ 0 & 1 & 0 & 3 \\ 0 & 0 & 1 & 5 \end{bmatrix}$$

Cramer's rule

```
> A1:=(submatrix(swapcol(Aa,1,4),1..3,1..3));
A2:=(submatrix(swapcol(Aa,2,4),1..3,1..3));
A3:=(submatrix(swapcol(Aa,3,4),1..3,1..3));
```

$$A1 := \begin{bmatrix} 0 & -3 & 1 \\ 3 & 2 & -1 \\ 12 & 1 & 1 \end{bmatrix}$$

$$A2 := \begin{bmatrix} 2 & 0 & 1 \\ 1 & 3 & -1 \\ 2 & 12 & 1 \end{bmatrix}$$

$$A3 := \begin{bmatrix} 2 & -3 & 0 \\ 1 & 2 & 3 \\ 2 & 1 & 12 \end{bmatrix}$$

```
> det(A1)/det(A); det(A2)/det(A); det(A3)/det(A);
```

$$2$$

$$3$$

$$5$$

System of linear equations as a matrix equation

```
> linsolve(A,B);
```

$$[2, 3, 5]$$

There are various tutorial commands to help students to learn some basic computations of linear algebra and calculus in the Student package.

```
> with(Student[LinearAlgebra]):
Warning, the name GramSchmidt has been rebound

Warning, the protected name . has been redefined and unprotected

> GaussJordanEliminationTutor(Matrix(A),B);
Initializing Java runtime environment.
```

EXAMPLE 3: Solve a system of linear equations which is specified by equations

```
> restart;
> r1:=x1+x2+x3+x4=0; r2:=x1+2*x2+3*x3+4*x4=0;
r3:=x1+3*x2+5*x3+7*x4=0; r4:=x1+4*x2+7*x3+10*x4=0;
```

$$r1 := x1 + x2 + x3 + x4 = 0$$

$$r2 := x1 + 2\,x2 + 3\,x3 + 4\,x4 = 0$$

$$r3 := x1 + 3\,x2 + 5\,x3 + 7\,x4 = 0$$

$$r4 := x1 + 4\,x2 + 7\,x3 + 10\,x4 = 0$$

```
> solve({r1,r2,r3,r4},{x1,x2,x3,x4});
```

$$\{x2 = -2\,x3 - 3\,x4,\ x1 = x3 + 2\,x4,\ x3 = x3,\ x4 = x4\}$$

```
> A:=linalg[genmatrix]({r1,r2,r3,r4},[x1,x2,x3,x4],'B');
```

$$A := \begin{bmatrix} 1 & 4 & 7 & 10 \\ 1 & 1 & 1 & 1 \\ 1 & 3 & 5 & 7 \\ 1 & 2 & 3 & 4 \end{bmatrix}$$

```
> LinearAlgebra[LinearSolve](Matrix(A),Vector(B),free='t');
```

$$\begin{bmatrix} t_3 + 2\,t_4 \\ -2\,t_3 - 3\,t_4 \\ t_3 \\ t_4 \end{bmatrix}$$

```
> linalg[geneqns](A,[x,y,z,v],B);
```

$$\{x + 4\,y + 7\,z + 10\,v = 0,\ x + y + z + v = 0,\ x + 3\,y + 5\,z + 7\,v = 0,\ x + 2\,y + 3\,z + 4\,v = 0\}$$

EXAMPLE 4: Find all real solutions to the system of equations. Give a graphic representation of the system of equations and its solution.

```
> restart;
> r1:=x+3*y+z=5; r2:=2*x+y+z=2; r3:=x+y+5*z=-7;
```

$$r1 := x + 3\,y + z = 5$$

$$r2 := 2\,x + y + z = 2$$

$$r3 := x + y + 5\,z = -7$$

```
> solve({r1,r2,r3},{x,y,z});
```

$$\{y = 2, x = 1, z = -2\}$$

```
> Student[LinearAlgebra][LinearSystemPlot]({r1,r2,r3});
```




```
> with(plots):
Warning, the name changecoords has been redefined

> R1:=implicitplot3d(r1,x=-5..5,y=-5..5,z=-
5..5,color=red,style=patchnogrid):
> R2:=implicitplot3d(r2,x=-5..5,y=-5..5,z=-
5..5,color=blue,style=patchnogrid):
> R3:=implicitplot3d(r3,x=-5..5,y=-5..5,z=-
5..5,color=green,style=patchnogrid):
```

```
> display(R1,R2,R3,axes=box);
```